In [1]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
import MeshCat as mc
```

```
  Activating environment at `~/ocrl_ws/16745-ocrl/HW3_S23/Project.toml`
Precompiling project...
  ✓ ConstructionBase
  ✓ NBInclude
  ✓ Mustache
  ✓ Setfield
  ✓ FiniteDiff
  ✓ Blink
  ✓ MeshCat
  ✓ MathOptInterface
  ✓ ECOS
  ✓ Convex
  ✓ Plots
  ✓ Ipopt
  12 dependencies successfully precompiled in 77 seconds (183 already precompiled)
```

In [2]:
```
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","cartpole_animation.jl"))
```

Out[2]:
```
animate_cartpole (generic function with 1 method)
```

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q1: Direct Collocation (DIRCOL) for a Cart Pole (30 pts)

We are now going to start working with the NonLinear Program (NLP) Solver IPOPT to solve some trajectory optimization problems. First we will demonstrate how this works for simple optimization problems (not trajectory optimization). The interface that we have setup for IPOPT is the following:

$$\begin{aligned}
\min_{x} \quad & \ell(x) && \text{cost function} && (1)\\
\text{st} \quad & c_{eq}(x) = 0 && \text{equality constraint} && (2)\\
& c_L \leq c_{ineq}(x) \leq c_U && \text{inequality constraint} && (3)\\
& x_L \leq x \leq x_U && \text{primal bound constraint} && (4)
\end{aligned}$$

where $\ell(x)$ is our objective function, $c_{eq}(x) = 0$ is our equality constraint, $c_L \leq c_{ineq}(x) \leq c_U$ is our bound inequality constraint, and $x_L \leq x \leq x_U$ is a bound constraint on our primal variable x.

Part A: Solve an LP with IPOPT (5 pts)

To demonstrate this, we are going to ask you to solve a simple Linear Program (LP):

$$\begin{aligned}
\min_{x} \quad & q^T x && (5)\\
\text{st} \quad & Ax = b && (6)\\
& Gx \leq h && (7)
\end{aligned}$$

Your job will be to transform this problem into the form shown above and solve it with IPOPT. To help you interface with IPOPT, we have created a function `fmincon` for you. Below is the docstring for this function that details all of the inputs.

In [3]:
```
"""
x = fmincon(cost,equality_constraint,inequality_constraint,x_l,x_u,c_l,c_u,x0,params,diff_type)

This function uses IPOPT to minimize an objective function

`cost(params, x)`

With the following three constraints:

`equality_constraint(params, x) = 0`
`c_l <= inequality_constraint(params, x) <= c_u`
`x_l <= x <= x_u`
```

```
Note that the constraint functions should return vectors.

Problem specific parameters should be loaded into params::NamedTuple (things like
cost weights, dynamics parameters, etc.).

args:
    cost::Function                    - objective function to be minimzed (returns scalar)
    equality_constraint::Function     - c_eq(params, x) == 0
    inequality_constraint::Function   - c_l <= c_ineq(params, x) <= c_u
    x_l::Vector                       - x_l <= x <= x_u
    x_u::Vector                       - x_l <= x <= x_u
    c_l::Vector                       - c_l <= c_ineq(params, x) <= x_u
    c_u::Vector                       - c_l <= c_ineq(params, x) <= x_u
    x0::Vector                        - initial guess
    params::NamedTuple                - problem parameters for use in costs/constraints
    diff_type::Symbol                 - :auto for ForwardDiff, :finite for FiniteDiff
    verbose::Bool                     - true for IPOPT output, false for nothing

optional args:
    tol                               - optimality tolerance
    c_tol                             - constraint violation tolerance
    max_iters                         - max iterations
    verbose                           - verbosity of IPOPT

outputs:
    x::Vector                         - solution

You should try and use :auto for your `diff_type` first, and only use :finite if you
absolutely cannot get ForwardDiff to work.

This function will run a few basic checks before sending the problem off to IPOPT to
solve. The outputs of these checks will be reported as the following:

---------checking dimensions of everything----------
---------all dimensions good-----------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------

If you're getting stuck during the testing of one of the derivatives, try switching
to FiniteDiff.jl by setting diff_type = :finite.
""";
```

In [9]:

```julia
@testset "solve LP with IPOPT" begin

    LP = jldopen(joinpath(@__DIR__,"utils","random_LP.jld2"))

    params = (q = LP["q"], A = LP["A"], b = LP["b"], G = LP["G"], h = LP["h"])

    # return a scalar
    function cost(params, x)::Real
        # create cost function with params and x
        return params.q' * x

    end

    # return a vector
    function equality_constraint(params, x)::Vector
        # create equality constraint function with params and x
        return params.A * x - params.b
    end

    # return a vector
    function inequality_constraint(params, x)::Vector
        # create inequality constraint function with params and x
        return params.G * x - params.h
    end

    # TODO: primal bounds
    # you may use Inf, like Inf*ones(10) for a vector of positive infinity
    x_l = -Inf * ones(20)
    x_u = Inf * ones(20)

    # TODO: inequality constraint bounds
    c_l = -Inf * ones(20)
    c_u = 0 * ones(20)

    # initial guess
    x0 = randn(20)

    diff_type = :auto   # use ForwardDiff.jl
#     diff_type = :finite # use FiniteDiff.jl

    x = fmincon(cost, equality_constraint, inequality_constraint,
```

```
                x_l, x_u, c_l, c_u, x0, params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true);


    @test isapprox(x, [-0.44289, 0, 0, 0.19214, 0, 0, -0.109095,
                       -0.43221, 0, 0, 0.44289, 0, 0, 0.192142,
                       0, 0, 0.10909, 0.432219, 0, 0], atol = 1e-3)

end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve---------------------

******************************************************************************
This program contains Ipopt, a library for large-scale nonlinear optimization.
 Ipopt is released as open source code under the Eclipse Public License (EPL).
         For more information visit https://github.com/coin-or/Ipopt
******************************************************************************

This is Ipopt version 3.14.4, running with linear solver MUMPS 5.4.1.

Number of nonzeros in equality constraint Jacobian...:       80
Number of nonzeros in inequality constraint Jacobian.:      400
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:       20
                     variables with only lower bounds:        0
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:        4
Total number of inequality constraints...............:       20
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:       20

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0 -3.5497588e+00 3.99e+00 3.33e-01   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  1.1648032e+00 2.76e+00 2.80e+00   0.1 3.35e+00    -  8.67e-02 3.08e-01f  1
   2  5.1867491e+00 3.33e-16 9.96e-07  -0.8 1.59e+00    -  1.00e+00 1.00e+00h  1
   3  1.7787522e+00 5.55e-17 2.28e-07  -1.5 6.69e-01    -  1.00e+00 9.62e-01f  1
   4  1.3324950e+00 5.55e-17 7.14e-08  -2.4 1.75e-01    -  7.70e-01 8.62e-01f  1
   5  1.1913173e+00 2.22e-16 3.06e-09  -3.4 8.26e-02    -  9.98e-01 9.14e-01f  1
   6  1.1766151e+00 5.55e-17 1.03e-10  -4.9 9.28e-03    -  1.00e+00 9.97e-01f  1
   7  1.1763494e+00 2.22e-16 3.16e-13 -10.8 1.23e-04    -  1.00e+00 1.00e+00f  1

Number of Iterations....: 7

                                   (scaled)                 (unscaled)
Objective...............:   1.1763493976085779e+00    1.1763493976085779e+00
Dual infeasibility......:   3.1619151741324458e-13    3.1619151741324458e-13
Constraint violation....:   2.2204460492503131e-16    2.2204460492503131e-16
Variable bound violation:   0.0000000000000000e+00    0.0000000000000000e+00
Complementarity.........:   9.4747618955315081e-09    9.4747618955315081e-09
Overall NLP error.......:   9.4747618955315081e-09    9.4747618955315081e-09


Number of objective function evaluations             = 8
Number of objective gradient evaluations             = 8
Number of equality constraint evaluations            = 8
Number of inequality constraint evaluations          = 8
Number of equality constraint Jacobian evaluations   = 8
Number of inequality constraint Jacobian evaluations = 8
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 1.797

EXIT: Optimal Solution Found.
Test Summary:      | Pass  Total
solve LP with IPOPT |    1      1
```

Out[9]:
```
Test.DefaultTestSet("solve LP with IPOPT", Any[], 1, false, false)
```

Part B: Cart Pole Swingup (20 pts)

We are now going to solve for a cartpole swingup. The state for the cartpole is the following:

$$x = [p, \theta, \dot{p}, \dot{\theta}]^T$$

Where p and θ can be seen in the graphic `cartpole.png`.

where we start with the pole in the down position ($\theta = 0$), and we want to use the horizontal force on the cart to drive the pole to the up position ($\theta = \pi$).

$$
\begin{aligned}
\min_{x_{1:N},u_{1:N-1}} \quad & \sum_{i=1}^{N-1}\left[\frac{1}{2}(x_i - x_{goal})^T Q (x_i - x_{goal}) + \frac{1}{2}u_i^T R u_i\right] + \frac{1}{2}(x_N - x_{goal})^T Q_f (x_N - x_{goal}) && (8)\\
\text{st} \quad & x_1 = x_{\text{IC}} && (9)\\
& x_N = x_{goal} && (10)\\
& f_{hs}(x_i, x_{i+1}, u_i, dt) = 0 \quad \text{for } i = 1,2,\ldots,N-1 && (11)\\
& -10 \leq u_i \leq 10 \quad \text{for } i = 1,2,\ldots,N-1 && (12)
\end{aligned}
$$

Where $x_{IC} = [0, 0, 0, 0]$, and $x_{goal} = [0, \pi, 0, 0]$, and $f_{hs}(x_i, x_{i+1}, u_i)$ is the implicit integrator residual for Hermite Simpson (see HW1Q1 to refresh on this). Note that while Zac used a first order hold (FOH) on the controls in class (meaning we linearly interpolate controls between time steps), we are using a zero-order hold (ZOH) in this assignment. This means that each control u_i is held constant for the entirety of the timestep.

In [14]:
```
# cartpole
function dynamics(params::NamedTuple, x::Vector, u)
    # cartpole ODE, parametrized by params.

    # cartpole physical parameters
    mc, mp, l = params.mc, params.mp, params.l
    g = 9.81

    q = x[1:2]
    qd = x[3:4]

    s = sin(q[2])
    c = cos(q[2])

    H = [mc+mp mp*l*c; mp*l*c mp*l^2]
    C = [0 -mp*qd[2]*l*s; 0 0]
    G = [0, mp*g*l*s]
    B = [1, 0]

    qdd = -H\(C*qd + G - B*u[1])
    xdot = [qd;qdd]
    return xdot

end
function hermite_simpson(params::NamedTuple, x1::Vector, x2::Vector, u, dt::Real)::Vector
    # input hermite simpson implicit integrator residual

    xk_dot = dynamics(params, x1, u)
    xk1_dot = dynamics(params, x2, u)

    xk_half = 0.5 * (x1 + x2) + dt .* (xk_dot - xk1_dot) / 8
    xk_half_dot = dynamics(params, xk_half, u)

    return x1 + dt .* (xk_dot + 4 * xk_half_dot + xk1_dot) / 6 - x2
end
```

Out[14]:
```
hermite_simpson (generic function with 1 method)
```

To solve this problem with IPOPT and `fmincon`, we are going to concatenate all of our x's and u's into one vector:

$$
Z = \begin{bmatrix} x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N\cdot nx + (N-1)\cdot nu}
$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z.

It is also worth noting that while there are inequality constraints present ($-10 \leq u_i \leq 10$), we do not need a specific `inequality_constraints` function as an input to `fmincon` since these are just bounds on the primal (Z) variable. You should use primal bounds in `fmincon` to capture these constraints.

In [47]:
```
function create_idx(nx,nu,N)
    # This function creates some useful indexing tools for Z
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]

    # Feel free to use/not use anything here.
```

```
    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function cartpole_cost(params::NamedTuple, Z::Vector)::Real
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf

    # input cartpole LQR cost

    J = 0
    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]

        J += 0.5 * (xi - xg)' * Q * (xi - xg) + 0.5 * ui' * R * ui
    end

    # dont forget terminal cost
    J += 0.5 * (Z[idx.x[N]] - xg)' * Q * (Z[idx.x[N]] - xg)

    return J
end

function cartpole_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt

    # create dynamics constraints using hermite simpson

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)

    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]
        xip1 = Z[idx.x[i+1]]

        # hermite simpson
        c[idx.c[i]] = hermite_simpson(params, xi, xip1, ui, dt)
    end
    return c
end

function cartpole_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # return all of the equality constraints

    return [Z[idx.x[1]] - xic;
            Z[idx.x[N]] - xg;
            cartpole_dynamics_constraints(params, Z)]
#     return cartpole_dynamics_constraints(params, Z)
end

function solve_cartpole_swingup(;verbose=true)

    # problem size
    nx = 4
    nu = 1
    dt = 0.05
    tf = 2.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # LQR cost
    Q = diagm(ones(nx))
    R = 0.1*diagm(ones(nu))
    Qf = 10*diagm(ones(nx))

    # indexing
    idx = create_idx(nx,nu,N)

    # initial and goal states
    xic = [0, 0, 0, 0]
    xg = [0, pi, 0, 0]

    # load all useful things into params
    params = (Q = Q, R = R, Qf = Qf, xic = xic, xg = xg, dt = dt, N = N, idx = idx,mc = 1.0, mp = 0.2, l = 0.5)
```

```
    # TODO: primal bounds
    x_l = -Inf * ones(idx.nz)
    x_u = Inf * ones(idx.nz)
    for i = 1:N-1
        x_l[idx.u[i]] .= -10
        x_u[idx.u[i]] .= 10
    end

    # Question - Ask TA where to specify Ui bounds

    # inequality constraint bounds (this is what we do when we have no inequality constraints)
    c_l = zeros(0)
    c_u = zeros(0)
    function inequality_constraint(params, Z)
        return zeros(eltype(Z), 0)
    end

    # initial guess
    z0 = 0.001*randn(idx.nz)

    # choose diff type (try :auto, then use :finite if :auto doesn't work)
    diff_type = :auto
#     diff_type = :finite

    Z = fmincon(cartpole_cost,cartpole_equality_constraint,inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = verbose)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    return X, U, t_vec, params
end

@testset "cartpole swingup" begin

    X, U, t_vec = solve_cartpole_swingup(verbose=true)


    # --------------testing------------------
    @test isapprox(X[1],zeros(4), atol = 1e-4)
    @test isapprox(X[end], [0,pi,0,0], atol = 1e-4)
    Xm = hcat(X...)
    Um = hcat(U...)

    # --------------plotting-----------------
    display(plot(t_vec, Xm', label = ["p" "θ" "ṗ" "θ̇"], xlabel = "time (s)", title = "State Trajectory"))
    display(plot(t_vec[1:end-1],Um',label="",xlabel = "time (s)", ylabel = "u",title = "Controls"))

    # meshcat animation
    display(animate_cartpole(X, 0.05))

end
```

```
---------checking dimensions of everything----------
---------all dimensions good-----------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------
This is Ipopt version 3.14.4, running with linear solver MUMPS 5.4.1.

Number of nonzeros in equality constraint Jacobian...:    34272
Number of nonzeros in inequality constraint Jacobian.:        0
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      204
                     variables with only lower bounds:        0
                variables with lower and upper bounds:       40
                     variables with only upper bounds:        0
Total number of equality constraints.................:      168
Total number of inequality constraints...............:        0
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.0229682e+02 3.14e+00 3.71e-04   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  2.4950063e+02 2.38e+00 7.96e+00  -5.0 1.28e+01    -  4.90e-01 2.43e-01h  3
   2  2.7667163e+02 2.16e+00 1.03e+01  -0.5 1.06e+01    -  6.09e-01 9.17e-02h  4
   3  3.1763757e+02 1.87e+00 1.41e+01  -0.4 1.30e+01    -  6.39e-01 1.32e-01h  3
   4  3.6124464e+02 1.60e+00 2.14e+01  -0.4 1.17e+01    -  9.77e-01 1.47e-01h  3
   5  4.1662392e+02 1.31e+00 2.77e+01  -0.7 9.66e+00    -  1.00e+00 1.78e-01h  3
   6  4.4488453e+02 1.18e+00 3.28e+01  -0.1 1.71e+01    -  4.75e-01 9.98e-02h  3
   7  4.7545220e+02 1.07e+00 3.63e+01   0.4 1.79e+01    -  5.03e-01 9.89e-02h  3
   8  5.1174496e+02 9.48e-01 4.03e+01   0.5 2.21e+01    -  9.77e-01 1.10e-01h  3
   9  5.2237246e+02 8.54e-01 3.78e+01   0.3 1.15e+01    -  5.99e-01 9.91e-02h  3
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  5.1776634e+02 7.64e-01 3.90e+01   0.5 2.53e+01    -  7.91e-01 1.05e-01f  3
  11  5.1634392e+02 6.82e-01 4.11e+01   0.6 2.04e+01    -  5.78e-01 1.08e-01h  3
  12  5.1785327e+02 6.00e-01 4.24e+01   0.5 2.43e+01    -  7.39e-01 1.20e-01h  3
  13  5.4794540e+02 3.45e-01 1.08e+02   0.7 1.89e+01    -  2.65e-01 4.26e-01H  1
  14  5.4834664e+02 3.15e-01 1.05e+02  -5.4 1.57e+01    -  2.89e-01 8.49e-02h  1
  15  5.5522176e+02 2.23e-01 9.21e+01   0.7 1.62e+01    -  4.98e-01 2.93e-01h  1
  16  5.5862760e+02 1.73e-01 9.24e+01   0.9 1.44e+01    -  7.40e-01 2.24e-01h  1
  17  5.5572361e+02 1.13e-01 1.10e+02   0.5 1.29e+01    -  4.89e-01 3.45e-01h  1
  18  5.5637088e+02 9.40e-02 1.01e+02   0.7 9.97e+00    -  9.36e-01 4.93e-01h  1
  19  5.2777318e+02 3.00e-02 7.96e+01   0.5 7.60e+00    -  8.67e-01 9.94e-01F  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  4.9624250e+02 2.41e-01 9.66e+01   0.5 1.87e+01    -  4.58e-01 3.88e-01f  1
  21  4.7323353e+02 9.96e-02 5.34e+01  -0.1 8.80e+00    -  3.89e-01 6.45e-01f  1
  22  4.5955679e+02 5.82e-02 2.74e+01  -1.4 6.38e+00    -  3.98e-01 6.08e-01f  1
  23  4.5528051e+02 2.70e-02 2.18e+01  -0.1 6.36e+00    -  8.29e-01 1.00e+00f  1
  24  4.4786664e+02 6.75e-02 3.88e+01  -0.3 5.86e+00    -  9.86e-01 1.00e+00f  1
  25  4.4211043e+02 1.07e-02 2.77e+01  -0.3 2.94e+00    -  8.94e-01 1.00e+00f  1
  26  4.3388882e+02 1.74e-02 3.99e+01  -0.0 1.09e+01    -  1.00e+00 9.37e-01F  1
  27  4.3336682e+02 9.36e-03 3.96e+01  -0.2 1.35e+01    -  1.00e+00 8.08e-01H  1
  28  4.2777215e+02 8.56e-03 2.48e+01  -0.4 3.20e+00    -  1.00e+00 1.00e+00f  1
  29  4.2489879e+02 3.28e-03 1.94e+01  -0.9 9.18e-01    -  9.97e-01 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  4.2321022e+02 1.85e-02 1.33e+01  -1.4 4.33e+00    -  1.00e+00 7.23e-01f  1
  31  4.2332437e+02 4.08e-02 2.06e+01  -0.8 1.73e+01    -  1.00e+00 2.14e-01f  2
  32  4.1971549e+02 1.02e-02 1.87e+01  -0.9 1.11e+01    -  4.38e-01 9.54e-01F  1
  33  4.1848482e+02 1.34e-02 2.56e+01  -1.1 3.06e+00    -  1.00e+00 1.00e+00f  1
  34  4.1806590e+02 2.38e-03 1.85e+01  -0.8 1.54e+00    -  1.00e+00 1.00e+00f  1
  35  4.1444685e+02 5.62e-03 1.92e+01  -1.4 5.31e+00    -  1.00e+00 1.00e+00F  1
  36  4.2097882e+02 1.18e-02 2.19e+01  -0.4 9.38e+00    -  8.93e-01 8.48e-01f  1
  37  4.1469484e+02 1.79e-02 1.87e+01  -0.5 2.77e+00    -  1.00e+00 1.00e+00f  1
  38  4.1382297e+02 2.76e-03 1.09e+01  -0.5 1.33e+00    -  8.81e-01 1.00e+00f  1
  39  4.1342396e+02 1.69e-02 1.71e+01  -0.7 1.82e+01    -  9.76e-01 1.45e-01f  3
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  40  4.0769850e+02 2.50e-02 2.53e+01  -0.4 1.01e+01    -  1.00e+00 1.00e+00F  1
  41  4.0917172e+02 6.49e-03 1.73e+01  -0.7 1.12e+00    -  9.97e-01 1.00e+00h  1
  42  4.0764041e+02 4.82e-03 1.26e+01  -1.6 1.10e+00    -  1.00e+00 1.00e+00f  1
  43  4.0144123e+02 1.88e-02 3.12e+01  -0.9 8.62e+00    -  8.70e-01 1.00e+00F  1
  44  4.0559599e+02 8.27e-03 1.70e+01  -0.7 4.47e+00    -  1.00e+00 1.00e+00h  1
  45  4.0359089e+02 4.98e-04 1.27e+01  -0.9 2.96e+00    -  9.97e-01 1.00e+00f  1
  46  4.0118078e+02 2.47e-03 1.44e+01  -1.6 2.44e+00    -  9.92e-01 1.00e+00F  1
  47  4.0307844e+02 1.12e-02 2.85e+01  -0.4 1.66e+01    -  9.35e-01 2.14e-01f  2
  48  4.0258573e+02 3.10e-02 3.39e+01  -0.5 4.95e+00    -  1.00e+00 1.00e+00f  1
  49  4.0021637e+02 1.10e-02 1.15e+01  -0.5 3.32e+00    -  9.50e-01 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  50  3.9801100e+02 1.76e-03 2.05e+01  -1.2 2.08e+00    -  9.93e-01 1.00e+00f  1
  51  3.9718495e+02 5.54e-04 1.29e+01  -1.4 1.22e+00    -  1.00e+00 1.00e+00f  1
  52  3.9663248e+02 6.55e-03 1.21e+01  -2.0 2.55e+00    -  1.00e+00 7.61e-01f  1
  53  4.0836765e+02 9.41e-03 1.96e+01  -0.8 1.33e+01    -  9.52e-01 1.00e+00H  1
  54  3.9361258e+02 3.78e-02 1.21e+01  -0.9 5.56e+00    -  1.00e+00 1.00e+00f  1
  55  3.9484883e+02 1.71e-03 4.46e+00  -1.4 9.09e-01    -  9.97e-01 1.00e+00h  1
  56  3.9455166e+02 2.01e-03 5.69e+00  -2.3 1.26e+00    -  1.00e+00 1.00e+00f  1
  57  3.9451543e+02 2.66e-03 8.20e+00  -1.7 7.43e+00    -  1.00e+00 9.96e-02f  3
```

```
  58  3.9417621e+02 6.70e-03 1.99e+01  -2.3 5.76e+00    -  1.00e+00 8.71e-01F  1
  59  3.9540713e+02 5.13e-03 1.37e+01  -1.9 3.04e+00    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  60  3.9350071e+02 3.55e-03 5.93e+00  -2.0 2.47e+00    -  1.00e+00 1.00e+00f  1
  61  3.9345997e+02 2.01e-05 2.17e+00  -3.1 1.65e-01    -  1.00e+00 1.00e+00h  1
  62  3.9344955e+02 9.46e-06 3.96e-01  -4.3 1.41e-01    -  1.00e+00 9.73e-01f  1
  63  3.9344844e+02 9.32e-07 4.11e-02  -5.9 4.31e-02    -  1.00e+00 1.00e+00h  1
  64  3.9344834e+02 5.40e-08 9.97e-04  -7.8 7.24e-03    -  1.00e+00 9.85e-01h  1
  65  3.9344834e+02 3.87e-08 1.92e-04  -9.5 4.95e-03    -  1.00e+00 1.00e+00h  1
  66  3.9344834e+02 5.42e-10 6.22e-05 -11.0 6.57e-04    -  1.00e+00 1.00e+00h  1
  67  3.9344834e+02 3.04e-10 8.59e-06 -11.0 5.30e-04    -  1.00e+00 1.00e+00h  1
  68  3.9344834e+02 1.56e-12 3.95e-06 -11.0 1.09e-04    -  1.00e+00 1.00e+00h  1
  69  3.9344834e+02 7.42e-13 2.73e-07 -11.0 3.67e-05    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 69

                                   (scaled)                 (unscaled)
Objective...............:   3.9344833576225585e+02    3.9344833576225585e+02
Dual infeasibility......:   2.7250028895582170e-07    2.7250028895582170e-07
Constraint violation....:   7.4162898044960457e-13    7.4162898044960457e-13
Variable bound violation:   9.9997231828297117e-08    9.9997231828297117e-08
Complementarity.........:   1.0000651959236603e-11    1.0000651959236603e-11
Overall NLP error.......:   2.7250028895582170e-07    2.7250028895582170e-07


Number of objective function evaluations             = 161
Number of objective gradient evaluations             = 70
Number of equality constraint evaluations            = 161
Number of inequality constraint evaluations          = 0
Number of equality constraint Jacobian evaluations   = 70
Number of inequality constraint Jacobian evaluations = 0
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 4.321

EXIT: Optimal Solution Found.
```






```
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
└ http://127.0.0.1:8701
```




```
Test Summary:    | Pass  Total
cartpole swingup |    2      2
```

Out[47]: Test.DefaultTestSet("cartpole swingup", Any[], 2, false, false)

Part C: Track DIRCOL Solution (5 pts)

Now, similar to HW2 Q2 Part C, we are taking a solution X and U from DIRCOL, and we are going to track the trajectory with TVLQR to account for model mismatch. While we used hermite-simpson integration for the dynamics constraints in DIRCOL, we are going to use RK4 for this simulation. Remember to clamp your control to be within the control bounds.

In [52]:
```
function rk4(params::NamedTuple, x::Vector,u,dt::Float64)
    # vanilla RK4
    k1 = dt*dynamics(params, x, u)
    k2 = dt*dynamics(params, x + k1/2, u)
    k3 = dt*dynamics(params, x + k2/2, u)
    k4 = dt*dynamics(params, x + k3, u)
    x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end

@testset "track cartpole swingup with TVLQR" begin

    X_dircol, U_dircol, t_vec, params_dircol = solve_cartpole_swingup(verbose = false)

    N = length(X_dircol)
    dt = params_dircol.dt
    x0 = X_dircol[1]

    # TODO: use TVLQR to generate K's

    # use this for TVLQR tracking cost
    Q = diagm([1,1,.05,.1])
    Qf = 100*Q
    R = 0.01*diagm(ones(1))

    nx = 4
    nu = 1
    A = [zeros(nx,nx) for i = 1:N-1]
    B = [zeros(nx,nu) for i = 1:N-1]
    K = [zeros(nu,nx) for i = 1:N-1]
    P = [zeros(nx,nx) for i = 1:N]

    for i = 1:N-1
        A[i] = ForwardDiff.jacobian(dx->rk4(params_dircol, dx, U_dircol[i], dt), X_dircol[i])
        B[i] = ForwardDiff.jacobian(du->rk4(params_dircol, X_dircol[i], du, dt), U_dircol[i])
    end

    P[N] = deepcopy(Qf)

    # Ricatti
    for k = N-1:-1:1
        K[k] = (R + B[k]' * P[k+1] * B[k])\ (B[k]' * P[k+1] * A[k])
        P[k] = Q + A[k]' * P[k+1] * (A[k] - B[k] * K[k])
    end
```

```
    # simulation
    Xsim = [zeros(4) for i = 1:N]
    Usim = [zeros(1) for i = 1:(N-1)]
    Xsim[1] = 1*x0

    # here are the real parameters (different than the one we used for DIRCOL)
    # this model mismatch is what's going to require the TVLQR controller to track
    # the trajectory successfully.
    params_real = (mc = 1.05, mp = 0.21, l = 0.48)

    # TODO: simulate closed loop system
    for i = 1:(N-1)
        # TODO: add feeback control (right now it's just feedforward)
        Usim[i] = clamp.(U_dircol[i] - K[i] * (Xsim[i] - X_dircol[i]), -10, 10) # update this
        Xsim[i+1] = rk4(params_real, Xsim[i], Usim[i], dt)
    end


    # -----------------testing----------------------
    xn = Xsim[N]
    @test norm(xn)>0
    @test 1e-6<norm(xn - X_dircol[end])<.8
    @test abs(abs(rad2deg(xn[2])) - 180) < 5 # within 5 degrees
    @test maximum(norm.(Usim,Inf)) <= (10 + 1e-3)

    # -----------------plotting---------------------
    Xm = hcat(Xsim...)
    Xbarm = hcat(X_dircol...)
    plot(t_vec,Xbarm',ls=:dash, label = "",lc = [:red :green :blue :black])
    display(plot!(t_vec,Xm',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "x",
                 label = ["p" "θ" "ṗ" "θ̇"],lc = [:red :green :blue :black]))

    Um = hcat(Usim...)
    Ubarm = hcat(U_dircol...)
    plot(t_vec[1:end-1],Ubarm',ls=:dash,lc = :blue, label = "")
    display(plot!(t_vec[1:end-1],Um',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "u",lc = :blue, label = ""))

    # ----------------animate-----------------------
    display(animate_cartpole(Xsim, 0.05))

end
```





┌ **Info:** MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
└ http://127.0.0.1:8705




```
Test Summary:                 | Pass  Total
track cartpole swingup with TVLQR |    4      4
```

Out[52]: `Test.DefaultTestSet("track cartpole swingup with TVLQR", Any[], 4, false, false)`

In [ ]: