In [10]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
import MeshCat as mc
```

```
  Activating environment at `~/ocrl_ws/16745-ocrl/HW3_S23/Project.toml`
```

In [11]:
```
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","cartpole_animation.jl"))
```

Out[11]:
```
animate_cartpole (generic function with 1 method)
```

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q1: Direct Collocation (DIRCOL) for a Cart Pole (30 pts)

We are now going to start working with the NonLinear Program (NLP) Solver IPOPT to solve some trajectory optimization problems. First we will demonstrate how this works for simple optimization problems (not trajectory optimization). The interface that we have setup for IPOPT is the following:

$$\begin{aligned} \min_{x} \quad & \ell(x) && \text{cost function} && (1) \\ \text{st} \quad & c_{eq}(x) = 0 && \text{equality constraint} && (2) \\ & c_L \leq c_{ineq}(x) \leq c_U && \text{inequality constraint} && (3) \\ & x_L \leq x \leq x_U && \text{primal bound constraint} && (4) \end{aligned}$$

where $\ell(x)$ is our objective function, $c_{eq}(x) = 0$ is our equality constraint, $c_L \leq c_{ineq}(x) \leq c_U$ is our bound inequality constraint, and $x_L \leq x \leq x_U$ is a bound constraint on our primal variable x.

Part A: Solve an LP with IPOPT (5 pts)

To demonstrate this, we are going to ask you to solve a simple Linear Program (LP):

$$\begin{aligned} \min_{x} \quad & q^T x && (5) \\ \text{st} \quad & Ax = b && (6) \\ & Gx \leq h && (7) \end{aligned}$$

Your job will be to transform this problem into the form shown above and solve it with IPOPT. To help you interface with IPOPT, we have created a function `fmincon` for you. Below is the docstring for this function that details all of the inputs.

In [12]:
```
"""
x = fmincon(cost,equality_constraint,inequality_constraint,x_l,x_u,c_l,c_u,x0,params,diff_type)

This function uses IPOPT to minimize an objective function

`cost(params, x)`

With the following three constraints:

`equality_constraint(params, x) = 0`
`c_l <= inequality_constraint(params, x) <= c_u`
`x_l <= x <= x_u`

Note that the constraint functions should return vectors.

Problem specific parameters should be loaded into params::NamedTuple (things like
cost weights, dynamics parameters, etc.).

args:
    cost::Function                    - objective function to be minimzed (returns scalar)
    equality_constraint::Function     - c_eq(params, x) == 0
    inequality_constraint::Function   - c_l <= c_ineq(params, x) <= c_u
    x_l::Vector                       - x_l <= x <= x_u
    x_u::Vector                       - x_l <= x <= x_u
    c_l::Vector                       - c_l <= c_ineq(params, x) <= x_u
    c_u::Vector                       - c_l <= c_ineq(params, x) <= x_u
```

```
    x0::Vector                          - initial guess
    params::NamedTuple                  - problem parameters for use in costs/constraints
    diff_type::Symbol                   - :auto for ForwardDiff, :finite for FiniteDiff
    verbose::Bool                       - true for IPOPT output, false for nothing

optional args:
    tol                                 - optimality tolerance
    c_tol                               - constraint violation tolerance
    max_iters                           - max iterations
    verbose                             - verbosity of IPOPT

outputs:
    x::Vector                           - solution

You should try and use :auto for your `diff_type` first, and only use :finite if you
absolutely cannot get ForwardDiff to work.

This function will run a few basic checks before sending the problem off to IPOPT to
solve. The outputs of these checks will be reported as the following:

---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve---------------------

If you're getting stuck during the testing of one of the derivatives, try switching
to FiniteDiff.jl by setting diff_type = :finite.
""";
```

In [13]:

```
@testset "solve LP with IPOPT" begin

    LP = jldopen(joinpath(@__DIR__,"utils","random_LP.jld2"))

    params = (q = LP["q"], A = LP["A"], b = LP["b"], G = LP["G"], h = LP["h"])

    # return a scalar
    function cost(params, x)::Real
        # create cost function with params and x
        return params.q' * x

    end

    # return a vector
    function equality_constraint(params, x)::Vector
        # create equality constraint function with params and x
        return params.A * x - params.b
    end

    # return a vector
    function inequality_constraint(params, x)::Vector
        # create inequality constraint function with params and x
        return params.G * x - params.h
    end

    # TODO: primal bounds
    # you may use Inf, like Inf*ones(10) for a vector of positive infinity
    x_l = -Inf * ones(20)
    x_u = Inf * ones(20)

    # TODO: inequality constraint bounds
    c_l = -Inf * ones(20)
    c_u = 0 * ones(20)

    # initial guess
    x0 = randn(20)

    diff_type = :auto   # use ForwardDiff.jl
#     diff_type = :finite # use FiniteDiff.jl

    x = fmincon(cost, equality_constraint, inequality_constraint,
                x_l, x_u, c_l, c_u, x0, params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true);


    @test isapprox(x, [-0.44289, 0, 0, 0.19214, 0, 0, -0.109095,
                       -0.43221, 0, 0, 0.44289, 0, 0, 0.192142,
                       0, 0, 0.10909, 0.432219, 0, 0], atol = 1e-3)

end
```

```
---------checking dimensions of everything----------
---------all dimensions good-----------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient----------------
---------testing constraint Jacobian---------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve---------------------
This is Ipopt version 3.14.4, running with linear solver MUMPS 5.4.1.

Number of nonzeros in equality constraint Jacobian...:       80
Number of nonzeros in inequality constraint Jacobian.:      400
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:       20
                     variables with only lower bounds:        0
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:        4
Total number of inequality constraints...............:       20
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:       20

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.3071134e+00 1.97e+00 3.33e-01   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  3.7831510e+00 6.92e-01 2.02e+00  -0.5 1.94e+00    -  2.22e-01 6.48e-01f  1
   2  3.1911447e+00 2.22e-16 7.86e-04  -1.6 1.06e+00    -  9.98e-01 1.00e+00h  1
   3  1.5717719e+00 2.22e-16 1.26e-07  -1.7 1.18e+00    -  1.00e+00 8.87e-01f  1
   4  1.2788550e+00 1.11e-16 1.33e-08  -2.7 9.23e-02    -  1.00e+00 7.91e-01f  1
   5  1.1797701e+00 5.55e-17 1.93e-09  -3.9 3.55e-02    -  9.16e-01 9.93e-01f  1
   6  1.1763544e+00 2.22e-16 1.40e-11  -9.7 2.15e-03    -  9.93e-01 9.99e-01f  1
   7  1.1763494e+00 2.22e-16 6.27e-15 -11.0 4.25e-06    -  1.00e+00 1.00e+00f  1

Number of Iterations....: 7

                                   (scaled)                 (unscaled)
Objective...............:   1.1763493513091967e+00    1.1763493513091967e+00
Dual infeasibility......:   6.2741891675300036e-15    6.2741891675300036e-15
Constraint violation....:   2.2204460492503131e-16    2.2204460492503131e-16
Variable bound violation:   0.0000000000000000e+00    0.0000000000000000e+00
Complementarity.........:   3.5916654841833766e-11    3.5916654841833766e-11
Overall NLP error.......:   3.5916654841833766e-11    3.5916654841833766e-11


Number of objective function evaluations             = 8
Number of objective gradient evaluations             = 8
Number of equality constraint evaluations            = 8
Number of inequality constraint evaluations          = 8
Number of equality constraint Jacobian evaluations   = 8
Number of inequality constraint Jacobian evaluations = 8
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 1.970

EXIT: Optimal Solution Found.
Test Summary:       | Pass  Total
solve LP with IPOPT |    1      1
```

Out[13]:
```
Test.DefaultTestSet("solve LP with IPOPT", Any[], 1, false, false)
```

Part B: Cart Pole Swingup (20 pts)

We are now going to solve for a cartpole swingup. The state for the cartpole is the following:

$$x = [p, \theta, \dot{p}, \dot{\theta}]^T$$

Where p and θ can be seen in the graphic `cartpole.png`.

where we start with the pole in the down position ($\theta = 0$), and we want to use the horizontal force on the cart to drive the pole to the up position ($\theta = \pi$).

$$\min_{x_{1:N}, u_{1:N-1}} \quad \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{goal})^T Q (x_i - x_{goal}) + \frac{1}{2} u_i^T R u_i \right] + \frac{1}{2}(x_N - x_{goal})^T Q_f (x_N - x_{goal}) \tag{8}$$

$$\text{st} \quad x_1 = x_{\text{IC}} \tag{9}$$

$$x_N = x_{goal} \tag{10}$$

$$f_{hs}(x_i, x_{i+1}, u_i, dt) = 0 \quad \text{for } i = 1, 2, \ldots, N-1 \tag{11}$$

$$-10 \leq u_i \leq 10 \quad \text{for } i = 1, 2, \ldots, N-1 \tag{12}$$

Where $x_{IC} = [0, 0, 0, 0]$, and $x_{goal} = [0, \pi, 0, 0]$, and $f_{hs}(x_i, x_{i+1}, u_i)$ is the implicit integrator residual for Hermite Simpson (see HW1Q1 to refresh on this). Note that while Zac used a first order hold (FOH) on the controls in class (meaning we linearly interpolate controls between time steps), we are using a zero-order hold (ZOH) in this assignment. This means that each control u_i is held constant for the entirety of the timestep.

In [14]:
```
# cartpole
function dynamics(params::NamedTuple, x::Vector, u)
    # cartpole ODE, parametrized by params.

    # cartpole physical parameters
    mc, mp, l = params.mc, params.mp, params.l
    g = 9.81

    q = x[1:2]
    qd = x[3:4]

    s = sin(q[2])
    c = cos(q[2])

    H = [mc+mp mp*l*c; mp*l*c mp*l^2]
    C = [0 -mp*qd[2]*l*s; 0 0]
    G = [0, mp*g*l*s]
    B = [1, 0]

    qdd = -H\(C*qd + G - B*u[1])
    xdot = [qd;qdd]
    return xdot

end
function hermite_simpson(params::NamedTuple, x1::Vector, x2::Vector, u, dt::Real)::Vector
    # input hermite simpson implicit integrator residual

    xk_dot = dynamics(params, x1, u)
    xk1_dot = dynamics(params, x2, u)

    xk_half = 0.5 * (x1 + x2) + dt .* (xk_dot - xk1_dot) / 8
    xk_half_dot = dynamics(params, xk_half, u)

    return x1 + dt .* (xk_dot + 4 * xk_half_dot + xk1_dot) / 6 - x2
end
```

Out[14]:
```
hermite_simpson (generic function with 1 method)
```

To solve this problem with IPOPT and `fmincon`, we are going to concatenate all of our x's and u's into one vector:

$$Z = \begin{bmatrix} x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N \cdot nx + (N-1) \cdot nu}$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z.

It is also worth noting that while there are inequality constraints present ($-10 \leq u_i \leq 10$), we do not need a specific `inequality_constraints` function as an input to `fmincon` since these are just bounds on the primal (Z) variable. You should use primal bounds in `fmincon` to capture these constraints.

In [15]:
```
function create_idx(nx,nu,N)
    # This function creates some useful indexing tools for Z
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]

    # Feel free to use/not use anything here.


    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function cartpole_cost(params::NamedTuple, Z::Vector)::Real
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf
```

```julia
    # input cartpole LQR cost

    J = 0
    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]

        J += 0.5 * (xi - xg)' * Q * (xi - xg) + 0.5 * ui' * R * ui
    end

    # dont forget terminal cost
    J += 0.5 * (Z[idx.x[N]] - xg)' * Qf * (Z[idx.x[N]] - xg)

    return J
end

function cartpole_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt

    # create dynamics constraints using hermite simpson

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)

    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]
        xip1 = Z[idx.x[i+1]]

        # hermite simpson
        c[idx.c[i]] = hermite_simpson(params, xi, xip1, ui, dt)
    end
    return c
end

function cartpole_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # return all of the equality constraints

    return [Z[idx.x[1]] - xic;
            Z[idx.x[N]] - xg;
            cartpole_dynamics_constraints(params, Z)]
#     return cartpole_dynamics_constraints(params, Z)
end

function solve_cartpole_swingup(;verbose=true)

    # problem size
    nx = 4
    nu = 1
    dt = 0.05
    tf = 2.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # LQR cost
    Q = diagm(ones(nx))
    R = 0.1*diagm(ones(nu))
    Qf = 10*diagm(ones(nx))

    # indexing
    idx = create_idx(nx,nu,N)

    # initial and goal states
    xic = [0, 0, 0, 0]
    xg = [0, pi, 0, 0]

    # load all useful things into params
    params = (Q = Q, R = R, Qf = Qf, xic = xic, xg = xg, dt = dt, N = N, idx = idx,mc = 1.0, mp = 0.2, l = 0.5)

    # TODO: primal bounds
    x_l = -Inf * ones(idx.nz)
    x_u = Inf * ones(idx.nz)
    for i = 1:N-1
        x_l[idx.u[i]] .= -10
        x_u[idx.u[i]] .= 10
    end

    # inequality constraint bounds (this is what we do when we have no inequality constraints)
    c_l = zeros(0)
    c_u = zeros(0)
    function inequality_constraint(params, Z)
        return zeros(eltype(Z), 0)
    end

    # initial guess
```

```
    z0 = 0.001*randn(idx.nz)

    # choose diff type (try :auto, then use :finite if :auto doesn't work)
    diff_type = :auto
#     diff_type = :finite

    Z = fmincon(cartpole_cost,cartpole_equality_constraint,inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = verbose)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    return X, U, t_vec, params
end

@testset "cartpole swingup" begin

    X, U, t_vec = solve_cartpole_swingup(verbose=true)


    # --------------testing------------------
    @test isapprox(X[1],zeros(4), atol = 1e-4)
    @test isapprox(X[end], [0,pi,0,0], atol = 1e-4)
    Xm = hcat(X...)
    Um = hcat(U...)

    # --------------plotting-----------------
    display(plot(t_vec, Xm', label = ["p" "θ" "ṗ" "θ̇"], xlabel = "time (s)", title = "State Trajectory"))
    display(plot(t_vec[1:end-1],Um',label="",xlabel = "time (s)", ylabel = "u",title = "Controls"))

    # meshcat animation
    display(animate_cartpole(X, 0.05))

end
```

```
---------checking dimensions of everything----------
---------all dimensions good-----------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------
This is Ipopt version 3.14.4, running with linear solver MUMPS 5.4.1.

Number of nonzeros in equality constraint Jacobian...:    34272
Number of nonzeros in inequality constraint Jacobian.:        0
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      204
                     variables with only lower bounds:        0
                variables with lower and upper bounds:       40
                     variables with only upper bounds:        0
Total number of equality constraints.................:      168
Total number of inequality constraints...............:        0
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.4668465e+02 3.14e+00 3.62e-04   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  2.7495449e+02 2.38e+00 7.99e+00  -5.0 1.28e+01    -  4.90e-01 2.43e-01h  3
   2  2.9800245e+02 2.16e+00 1.03e+01  -0.5 1.05e+01    -  6.11e-01 9.26e-02h  4
   3  3.3417253e+02 1.87e+00 1.40e+01  -0.4 1.29e+01    -  6.48e-01 1.33e-01h  3
   4  3.7111067e+02 1.61e+00 2.08e+01  -0.5 1.19e+01    -  8.79e-01 1.40e-01h  3
   5  4.1954974e+02 1.33e+00 2.73e+01  -0.8 1.00e+01    -  1.00e+00 1.74e-01h  3
   6  4.4373136e+02 1.20e+00 3.19e+01   0.3 1.84e+01    -  6.33e-01 9.61e-02h  3
   7  4.7556102e+02 1.07e+00 3.53e+01   0.2 1.80e+01    -  6.47e-01 1.12e-01h  3
   8  5.1167839e+02 9.43e-01 3.90e+01   0.3 2.25e+01    -  6.12e-01 1.17e-01h  3
   9  5.2133501e+02 8.53e-01 3.84e+01   0.3 1.15e+01    -  8.73e-01 9.51e-02h  3
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  5.1533080e+02 7.71e-01 4.12e+01   0.4 2.61e+01    -  5.18e-01 9.70e-02f  3
  11  5.0923158e+02 7.01e-01 4.40e+01   0.5 2.68e+01    -  5.97e-01 9.06e-02f  3
  12  5.0681641e+02 6.26e-01 4.91e+01   0.4 3.52e+01    -  8.46e-01 1.07e-01f  3
  13  5.0844912e+02 5.59e-01 5.71e+01   0.6 2.68e+01    -  3.33e-01 1.06e-01h  3
  14  5.3530304e+02 3.51e-01 7.08e+01   0.4 1.99e+01    -  1.94e-01 3.71e-01h  1
  15  5.3505331e+02 2.74e-01 7.29e+01   0.2 1.78e+01    -  2.68e-01 2.20e-01h  1
  16  5.4192440e+02 1.90e-01 7.58e+01   0.7 1.63e+01    -  3.88e-01 3.09e-01f  1
  17  5.3760775e+02 1.18e-01 8.54e+01   0.6 1.18e+01    -  7.66e-01 3.75e-01h  1
  18  5.4054375e+02 8.77e-02 7.83e+01   0.6 9.63e+00    -  8.00e-01 5.53e-01h  1
  19  5.2842831e+02 9.21e-02 5.78e+01   0.4 6.87e+00    -  8.15e-01 9.69e-01h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  5.0251932e+02 3.95e-02 2.13e+01   0.1 2.39e+00    -  9.71e-01 1.00e+00f  1
  21  4.8256354e+02 6.12e-02 2.47e+01   0.1 1.02e+01    -  5.27e-01 3.35e-01f  1
  22  4.6953842e+02 4.34e-02 1.77e+01  -0.1 5.15e+00    -  9.23e-01 4.65e-01f  1
  23  4.6319920e+02 2.27e-01 4.29e+01  -0.1 3.14e+01    -  3.04e-01 2.92e-01f  1
  24  4.6215449e+02 1.64e-01 5.91e+01   0.4 1.93e+01    -  1.00e+00 3.21e-01f  1
  25  4.4566372e+02 7.12e-03 2.84e+01  -0.1 2.84e+00    -  1.00e+00 1.00e+00f  1
  26  4.4068492e+02 5.58e-03 2.00e+01  -0.9 1.72e+00    -  9.96e-01 1.00e+00f  1
  27  4.3812554e+02 2.32e-03 1.89e+01  -1.5 1.35e+00    -  1.00e+00 1.00e+00f  1
  28  4.3723714e+02 2.85e-04 1.78e+01  -2.2 7.12e-01    -  1.00e+00 1.00e+00f  1
  29  4.3666209e+02 3.34e-03 1.85e+01  -2.8 1.38e+00    -  1.00e+00 9.97e-01f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  4.3686462e+02 4.71e-02 1.68e+01  -0.8 1.19e+01    -  2.16e-01 7.28e-01f  1
  31  4.3236676e+02 4.30e-02 1.37e+01  -1.4 4.47e+00    -  1.00e+00 1.00e+00f  1
  32  4.3293853e+02 2.43e-02 1.32e+01  -0.7 3.46e+00    -  1.00e+00 1.00e+00f  1
  33  4.3344139e+02 8.06e-02 2.31e+01  -0.4 2.44e+01    -  6.68e-01 2.72e-01f  2
  34  4.3320582e+02 8.96e-02 3.57e+01  -0.6 2.28e+01    -  4.53e-01 1.47e-01f  2
  35  4.2803343e+02 3.52e-02 3.12e+01  -0.6 3.91e+00    -  5.78e-01 8.66e-01f  1
  36  4.2177849e+02 3.01e-03 3.56e+01  -1.2 3.53e+00    -  1.00e+00 1.00e+00f  1
  37  4.2066409e+02 2.02e-02 1.95e+01  -1.2 5.10e+00    -  1.00e+00 9.57e-01f  1
  38  4.1990357e+02 9.71e-04 1.70e+01  -0.8 2.51e+00    -  1.00e+00 1.00e+00f  1
  39  4.1660440e+02 6.81e-03 2.17e+01  -1.3 4.03e+00    -  1.00e+00 1.00e+00F  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  40  4.1686813e+02 1.31e-02 2.95e+01  -0.6 1.05e+01    -  1.00e+00 2.98e-01f  2
  41  4.1403636e+02 1.06e-02 2.13e+01  -0.7 6.16e+00    -  1.00e+00 1.00e+00F  1
  42  4.1685841e+02 3.73e-03 1.71e+01  -0.5 1.80e+00    -  1.00e+00 1.00e+00h  1
  43  4.1502335e+02 4.08e-03 1.97e+01  -0.6 1.30e+00    -  1.00e+00 1.00e+00f  1
  44  4.1657785e+02 1.35e-03 1.85e+01  -0.3 3.65e+00    -  9.39e-01 1.00e+00F  1
  45  4.1136302e+02 1.23e-02 2.61e+01  -0.8 3.27e+00    -  9.99e-01 1.00e+00f  1
  46  4.0871210e+02 3.25e-03 1.73e+01  -1.2 2.58e+00    -  1.00e+00 1.00e+00f  1
  47  4.0852141e+02 4.87e-03 1.05e+01  -1.4 1.58e+00    -  9.98e-01 1.00e+00f  1
  48  4.0832794e+02 2.17e-03 1.43e+01  -1.0 2.01e+00    -  1.00e+00 1.00e+00f  1
  49  4.0565080e+02 1.04e-02 2.10e+01  -0.7 4.13e+00    -  1.00e+00 1.00e+00F  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  50  4.0555902e+02 9.68e-03 1.47e+01  -0.9 2.59e+00    -  9.97e-01 1.00e+00h  1
  51  4.0697803e+02 1.24e-02 1.13e+01  -0.1 1.02e+01    -  3.41e-01 1.64e-01f  2
  52  4.0675441e+02 3.36e-02 2.16e+01  -0.6 1.63e+01    -  5.51e-01 2.73e-01f  2
  53  4.0434740e+02 4.98e-03 1.83e+01  -0.6 5.95e+00    -  6.79e-01 1.00e+00F  1
  54  4.0192004e+02 2.34e-03 1.90e+01  -0.9 2.64e+00    -  1.00e+00 1.00e+00f  1
  55  4.0091123e+02 2.89e-04 1.82e+01  -1.5 7.27e-01    -  1.00e+00 1.00e+00f  1
  56  4.0051268e+02 1.11e-03 1.81e+01  -2.5 8.99e-01    -  1.00e+00 1.00e+00f  1
  57  4.0021069e+02 1.28e-03 2.24e+01  -1.7 3.19e+00    -  1.00e+00 3.16e-01f  1
```

```
  58  4.0019378e+02 2.04e-02 2.79e+01  -1.8 4.40e+00    -  1.00e+00 1.00e+00f  1
  59  4.0069332e+02 2.07e-02 1.75e+01  -0.9 5.42e+00    -  9.82e-01 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  60  3.9945861e+02 2.81e-02 2.97e+01  -1.0 4.71e+00    -  1.00e+00 1.00e+00f  1
  61  4.0246664e+02 7.24e-03 7.89e+00  -0.4 5.33e+00    -  9.86e-01 1.00e+00f  1
  62  3.9970069e+02 2.26e-02 1.64e+01  -0.5 3.89e+00    -  1.00e+00 1.00e+00f  1
  63  3.9690108e+02 1.32e-03 1.20e+01  -0.5 1.23e+00    -  1.00e+00 1.00e+00f  1
  64  3.9530030e+02 2.15e-03 2.76e+00  -1.2 1.31e+00    -  9.91e-01 1.00e+00f  1
  65  3.9484971e+02 3.89e-03 7.55e+00  -2.2 1.81e+00    -  1.00e+00 1.00e+00f  1
  66  3.9487685e+02 4.59e-03 1.95e+01  -2.8 4.22e+00    -  9.90e-01 1.00e+00H  1
  67  3.9409418e+02 2.42e-03 1.42e+01  -2.1 1.37e+00    -  1.00e+00 1.00e+00f  1
  68  3.9371712e+02 8.90e-05 8.32e+00  -2.3 5.61e-01    -  1.00e+00 1.00e+00f  1
  69  3.9364013e+02 1.52e-03 7.17e+00  -2.8 1.84e+00    -  1.00e+00 5.90e-01f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  70  3.9346987e+02 1.67e-04 8.15e-01  -2.8 6.82e-01    -  1.00e+00 1.00e+00f  1
  71  3.9344621e+02 4.45e-05 5.44e-01  -3.9 5.71e-01    -  1.00e+00 1.00e+00F  1
  72  3.9344921e+02 1.14e-06 3.02e-01  -4.3 2.28e-01    -  9.99e-01 1.00e+00H  1
  73  3.9344862e+02 1.11e-06 5.45e-02  -5.9 4.61e-02    -  1.00e+00 9.99e-01h  1
  74  3.9344840e+02 1.22e-06 1.79e-02  -7.0 3.33e-02    -  1.00e+00 1.00e+00h  1
  75  3.9344834e+02 6.46e-09 8.97e-05  -8.7 2.00e-03    -  1.00e+00 1.00e+00h  1
  76  3.9344834e+02 2.49e-10 6.33e-06 -11.0 4.50e-04    -  1.00e+00 1.00e+00h  1
  77  3.9344834e+02 2.64e-12 1.50e-06 -11.0 5.11e-05    -  1.00e+00 1.00e+00h  1
  78  3.9344834e+02 3.20e-14 3.79e-07 -11.0 1.50e-05    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 78

                                   (scaled)                 (unscaled)
Objective...............:   3.9344833576223027e+02    3.9344833576223027e+02
Dual infeasibility......:   3.7890749433606510e-07    3.7890749433606510e-07
Constraint violation....:   3.1974423109204508e-14    3.1974423109204508e-14
Variable bound violation:   9.9997231828297117e-08    9.9997231828297117e-08
Complementarity.........:   1.0000651658239830e-11    1.0000651658239830e-11
Overall NLP error.......:   3.7890749433606510e-07    3.7890749433606510e-07


Number of objective function evaluations             = 166
Number of objective gradient evaluations             = 79
Number of equality constraint evaluations            = 166
Number of inequality constraint evaluations          = 0
Number of equality constraint Jacobian evaluations   = 79
Number of inequality constraint Jacobian evaluations = 0
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 10.899

EXIT: Optimal Solution Found.
```






```
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
└ http://127.0.0.1:8716
```




```
Test Summary:    | Pass  Total
cartpole swingup |    2      2
```

Out[15]: `Test.DefaultTestSet("cartpole swingup", Any[], 2, false, false)`

Part C: Track DIRCOL Solution (5 pts)

Now, similar to HW2 Q2 Part C, we are taking a solution X and U from DIRCOL, and we are going to track the trajectory with TVLQR to account for model mismatch. While we used hermite-simpson integration for the dynamics constraints in DIRCOL, we are going to use RK4 for this simulation. Remember to clamp your control to be within the control bounds.

In [16]:
```
function rk4(params::NamedTuple, x::Vector,u,dt::Float64)
    # vanilla RK4
    k1 = dt*dynamics(params, x, u)
    k2 = dt*dynamics(params, x + k1/2, u)
    k3 = dt*dynamics(params, x + k2/2, u)
    k4 = dt*dynamics(params, x + k3, u)
    x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end

@testset "track cartpole swingup with TVLQR" begin

    X_dircol, U_dircol, t_vec, params_dircol = solve_cartpole_swingup(verbose = false)

    N = length(X_dircol)
    dt = params_dircol.dt
    x0 = X_dircol[1]

    # TODO: use TVLQR to generate K's
```

```julia
    # use this for TVLQR tracking cost
    Q = diagm([1,1,.05,.1])
    Qf = 100*Q
    R = 0.01*diagm(ones(1))

    nx = 4
    nu = 1
    A = [zeros(nx,nx) for i = 1:N-1]
    B = [zeros(nx,nu) for i = 1:N-1]
    K = [zeros(nu,nx) for i = 1:N-1]
    P = [zeros(nx,nx) for i = 1:N]

    for i = 1:N-1
        A[i] = ForwardDiff.jacobian(dx->rk4(params_dircol, dx, U_dircol[i], dt), X_dircol[i])
        B[i] = ForwardDiff.jacobian(du->rk4(params_dircol, X_dircol[i], du, dt), U_dircol[i])
    end

    P[N] = deepcopy(Qf)

    # Ricatti
    for k = N-1:-1:1
        K[k] = (R + B[k]' * P[k+1] * B[k])\ (B[k]' * P[k+1] * A[k])
        P[k] = Q + A[k]' * P[k+1] * (A[k] - B[k] * K[k])
    end

    # simulation
    Xsim = [zeros(4) for i = 1:N]
    Usim = [zeros(1) for i = 1:(N-1)]
    Xsim[1] = 1*x0

    # here are the real parameters (different than the one we used for DIRCOL)
    # this model mismatch is what's going to require the TVLQR controller to track
    # the trajectory successfully.
    params_real = (mc = 1.05, mp = 0.21, l = 0.48)

    # TODO: simulate closed loop system
    for i = 1:(N-1)
        # TODO: add feeback control (right now it's just feedforward)
        Usim[i] = clamp.(U_dircol[i] - K[i] * (Xsim[i] - X_dircol[i]), -10, 10) # update this
        Xsim[i+1] = rk4(params_real, Xsim[i], Usim[i], dt)
    end


    # -----------------testing----------------------
    xn = Xsim[N]
    @test norm(xn)>0
    @test 1e-6<norm(xn - X_dircol[end])<.8
    @test abs(abs(rad2deg(xn[2])) - 180) < 5 # within 5 degrees
    @test maximum(norm.(Usim,Inf)) <= (10 + 1e-3)

    # -----------------plotting---------------------
    Xm = hcat(Xsim...)
    Xbarm = hcat(X_dircol...)
    plot(t_vec,Xbarm',ls=:dash, label = "",lc = [:red :green :blue :black])
    display(plot!(t_vec,Xm',title = "Cartpole TVLQR (-- is reference)",
                  xlabel = "time (s)", ylabel = "x",
                  label = ["p" "θ" "ṗ" "θ̇"],lc = [:red :green :blue :black]))

    Um = hcat(Usim...)
    Ubarm = hcat(U_dircol...)
    plot(t_vec[1:end-1],Ubarm',ls=:dash,lc = :blue, label = "")
    display(plot!(t_vec[1:end-1],Um',title = "Cartpole TVLQR (-- is reference)",
                  xlabel = "time (s)", ylabel = "u",lc = :blue, label = ""))

    # ----------------animate-----------------------
    display(animate_cartpole(Xsim, 0.05))

end
```






```
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
└ http://127.0.0.1:8717
```




```
Test Summary:                      | Pass  Total
track cartpole swingup with TVLQR |    4      4
```

Out[16]:
```
Test.DefaultTestSet("track cartpole swingup with TVLQR", Any[], 4, false, false)
```